

漁況データ処理のためのパソコンプログラム

加藤 史彦

(日本海区水産研究所)

I ま え が き

漁業は一つの大規模な資源調査であり、そこで得られた情報は最大限に活用することが望ましい。漁況予報の分野においても漁獲統計は大きな働きをしている。昭和39年から始まった漁海況予報事業の一環として、毎日の地区別・漁業別・魚種別漁獲統計が収集されて、旬報または月報の形で公表されている。筆者はこれらのデータをファイル化してパソコンのディスクケットの中に保存し、必要に応じてひき出して加工して利用している。

近年パソコンの発達と普及はめざましく、それに呼応して表集計用の簡易言語も高性能のものが多数開発されている。こういった市販のソフトを利用するのも一つの方法であるが、その場合ファイルの構成がわからないのが最大の不安である。

ここに一例として掲げたプログラムは富山県の旬別データを保存するものであるが、これを日単位のファイルを作るように修正し、更に、このファイルを用いて月および年の合計を計算するプログラムを追加すれば、実用になると思われるので、あえて紹介することにした。

II 機器構成

1. パソコン本体：Hewlett・Packard モデル236C
プロセッサ：モトローラMC68000 (16ビット)
メモリ容量：640KBほかに256KBの拡張メモ리카ード増設
ディスプレイ：12インチカラーCRT文字画面80字×25桁
グラフィックス512×390ドット
補助記憶装置：5.25インチ (260KB) ディスクドライブ2基
2. 使用言語：HP拡張BASIC
3. プリンタ：HP2631Aドットマトリックスプリンタ 最大227桁/行

III プログラムの仕様

プログラムはメインプログラムと7つのサブルーチンとで構成されている。そのそれぞれについて以下に説明する。

1. メインプログラム

まず最初に変数をメモリーのコモン（共通）領域に確保する（1020-1060行）。行1110から1130および1200から1220ではそれに続く D A T A 文から列（縦）の項目名を読み込む。列の名前は魚種名などであるが長いものがあるので2行に表示する。したがって、1行目と2行目とは別の文字列変数 Item 1 \$, Item 2 \$ に入っている。行1300から1320および1410から1430行では行（横）の項目名である地区名 Chiku \$ と漁業種類名 Gyogyo \$ とをそれぞれ読み込む。行1570以下は必要なサブルーチンをソフトキー（ファンクションキー）を用いて呼ぶためのメニュールーチンである。

2. サブルーチン Mclear

漁業種類や地区はコード番号で魚種別漁獲量とともに配列変数 Datal に入力され、コモン領域に記憶されているので、次のファイルを作るとき必要に応じて変数の中身をご破算にする。MAT というステートメントが使えない場合は、FOR - NEXT 文を用いて配列の要素すべてをゼロにする。

3. サブルーチン Datin

データの入力はテレビ画面上の表を見ながら行う。PRINTER IS 1 というステートメントにより以後の PRINT 文による出力はすべて CRT 上になされる（行1720, 1730）。ついでサブルーチン Set を呼んでこれから作るファイルの年・月・旬を入力する。さらにサブルーチン Set 1 により表の何行目から何行目まで、または何桁目から何桁目までを入力するかを決定する。ここでは入力開始行を0とするとメニュー画面に戻るようになっている（行1770）。

CRT の文字画面の出力エリアは18行×80列しかないので表全体は一画面におさまらない。そこで表を縦方向に2つ、横方向に4つ、すなわち8分割して1画面とし表示することにした。分割した表の上半分は左から右に順番に11, 12, 13, 14と、下半分も同様に21, 22, 23, 24と Page 番号を割り振っている。そして入力した Start - gyo と Start - retsu とから Page 番号を計算し（行1780-1800）、サブルーチン Crt 1 を呼んで CRT の画面に表の指定のページを表示させる（行1820）。

データの入力は行（I）と列（J）で指定した（行1830, 1840）開始位置からスタートし、入力が進むと I と J の値は次第に変化するので、それに従って新しいページ番号 Page - next を計算し前の Page と違っていればサブルーチン Crt 1 をコールして新しいページを CRT に表示する（行1880-1940）。

HP 236 C ではカーソルは左右方向にしか動かず、上下方向の編集は画面をスクロールさせることによっておこなっている。したがって画面上の入力したい箇所を反転表示（インバース

モード) させてカーソルの代用に使っている (行1960, 1970, 1950-2050)。

漁業種類と地区はコード番号を, その他の漁獲量などのデータは実数をキイインし, 続いて CONTINUE キーを押すと数値が配列変数 Data 1 (I, J) の中に格納される。この時, データの代わりに-4, -6, -8 または-2 をキイインすると CONTINUE キーが押された後に, カーソルはそれぞれ左, 右, 上, および下に動く。また-5 の場合はカーソルはその場に停止し, -9 の時は入力終わりとみなして合計を計算するルーチン (Total) にとぶ (行2070-2310)。行2200から2230では標準モードで入力データを表示しカーソルを消している。

データとして-9 が入力された時に横の合計すなわち総漁獲量が計算される。合計値はデータ行列の38列目に入っているため, まず表の上半分のページ14を表示する。画面は PAUSE コマンドによって固定されるので, ここでサムチェックを行ったのちに CONTINUE キーを押すと, 下半分のページ24が表示されるようになっている (行2530-2700)。

4. サブルーチン Crt 1

表を CRT またはプリンタにアウトプットするサブルーチンであるが, 前者の場合は分割表が (行2770-2820), 後者の場合は表全体が出力される (行2830-2880)。HPBASIC には LPRINT というステートメントはなく, そのかわりに PRINTER IS の後に出力機器の番号 (CRT は 1, プリンタは 701) を与えた命令を実行すると, 以後の PRINT 文による出力はすべて指定した機器になされる (行2890)。また, PRINT USING の後にイメージ・ライン・ラベルを指定すると, そのラベルの IMAGE ステートメントに従った仕様のプリントができる (行2910-2920)。

印刷する表は横の桁数が多いので, プリンタを圧縮文字モードに変えて最大227桁まで印字できるようにする (行2940)。魚種名などの列の項目名は 1 行目は文字列変数 Item 1 \$ (J) に, 2 行目は Item 2 \$ (J) にそれぞれ入っており, これらは Format 1 \$ (J) のイメージ仕様にしたがってプリントする (行2950-3130)。1 列目の漁業種類と 2 列目の地区はコード番号で入力されているが, プリントアウトはそれに対応する文字列変数 Gyogyo\$, Chiku\$ をカタカナで行い, 地区および漁業のコード番号が許される範囲の 1-4 および 1-9 を越える場合には, 漁獲量と同様に入力値そのものを印刷する (行3140-3290)。

5. サブルーチン Set

データファイルの西暦年・月・旬を数値で入力し, VAL\$ 関数を用いて文字列に直し, ディスケットに保存するためのファイルの名前を合成する。上・中・下旬は 1, 2, 3 で入力するがファイル名では E, M, L を用いる。ファイル名は 7 桁よりなるが, 最初の 2 文字は富山県を示す TY で, 3, 4 文字目は西暦年の 10 の位と 1 の位を, 5, 6 文字目は月を, 7 文字目は

旬をあらわしている。

6. サブルーチン Output

プリンタにのみアウトプットするサブルーチンである。まずサブルーチン Set をコールして出力する表の年、月、旬を定め、次いで Set 1 をコールして表の何行目から何行目まで、または何桁目から何桁目までを印刷するかを決定する。最後に、 Crt 1 をコールしてプリントを行う。

7. サブルーチン Set 1

入力したり、出力したりする行や列を指定するサブルーチンである。行数と列数の最高はそれぞれ20と37であるので、それを越えると注意メッセージがでて、入れなおしをさせるようになっている。

8. サブルーチン Diskin

指定した1番（右側）の内蔵ディスクドライブから、サブルーチン Set で合成したファイル名のデータを読みとり、パソコンのメモリー上の配列変数 Data 1 (I, J) のなかに入れるルーチンである。

9. サブルーチン Diskout

入力したデータを1番のディスクドライブのディスケットに保存するサブルーチンである。保存用のファイルの名前はサブルーチン Set で作り、その名前のファイルがディスケットに既にあるか、否かをしらべ(行4400)、あればファイルの中身を書き変えることになるので注意メッセージを出す(行4410-4500)。ファイルがなければ新しいファイルをディスケット上に開いて、それにデータを出力する。1つのファイルには800個の実数型数値が保存できるように、8バイトのレコードを800確保してある。

10. サブルーチン Hp 9825 in

これは古い計算機 Hp 9825 A を用いて8インチのディスケットに保存してある同種のデータを、インターフェイスケーブル HPIB を介して HP 9825 A のメモリーから HP 236 C のメモリーに転送するサブルーチンである。

IV データの互換性

HP 236 C で作ったデータファイルを他の計算機で利用したり、異なる媒体に保存するこ

とは比較的容易である。それには、HP 236 C と他の計算機を直接 RS-232 C ケーブルでつなぐとよい。そしてディスク上のファイルから計算機のメモリーにデータを読みとり、RS-232 C を通して他の計算機にデータを転送し、必要に応じて他のディスクにおとすという作業を行えばよい。

このようにコンピュータどうしを直接むすんでデータを転送しようとするとき、気をつけねばならないことが一つある。それは、市販されている RS-232 C 用コードは音響ケーブルを使用することを標準としているので、両端のコネクタは対応するピン番号どうしが接続されており、2つのコンピュータの送信端子（2番ピン）及び受信端子（3番ピン）どうしがつながってしまうことである。従って、一方のコネクタを開けて、2番と3番の配線をつなぎかえ、同様に4番（送信要求）と5番（送信許可）についてもこの処置をほどこす必要がある。今では、RS-232 C の接続の切り替え装置も市販されているので、それも利用できる。

V む す び

データをファイル化して磁気媒体に保存する作業は大変手間もかかり、根気のいる仕事であるが、その際に入力ミスがあっては致命的である。ミスを防ぐには、なるべく調査表に近い形のファイルを作るのが良いと思われる。そして、このようにしてできあがったファイルが、将来有効にかつ効率的に使われることを望むものである。

付表-1 プログラムリスト

```

1000 ! ** トマケン キョクヨウ テーマ ファイル サクセイ ** TYGYOFILE3 ** June,1984 F.Kato ****
1010 ! ! の REM ト オナシ、 コメント テア。
1020 OPTION BASE 1 ! 0レツ ノ カケンヲ 1 ニ シテイス。
1030 COM Data1(20,38),Filename$(81),Item1$(38)[10],Item2$(38)[10],Chiku$(9)[8]
1040 COM Gyogyo$(4)[7],Format1$(38)[5],Format2$(38)[5]
1050 COM Page,Year,Month,Jun,Chiku,Start_gyop,End_gyop,Start_retsu,End_retsu
1060 COM Printer_no
1070 ! COM ニヨリ ハンズウ ヨ COMMON リョウイキ ニ カクホ、 SURROUTINE ニ ヒキワセ。
1080 ! アラマテ RUN シテム スウチ カ、 キエナイ。
1090 Filename$="TY0000E " ! テーマ ファイルメイ ヨ ショキカ ス。
1100 ! ----- キョクヨウ -----
1110 FOR I=1 TO 38
1120 READ Item1$(I)
1130 NEXT I
1140 DATA "キョクヨウ","チク","セキスウ","アシ","サハ","カタク","マイワシ"
1150 DATA "ウルメイシ","フリ","フクラキ","ツバイソ","ヒラマサ","マクロ","メシ"
1160 DATA "カワハキ","カマス","ソーダ","ニキス","シイラ","フク","チウオ","ホタルイカ"
1170 DATA "ヒラメ","タイ","アライカ","スゲトウタラ","サヨリ","サケ","ヘビス","アラ"
1180 DATA "オキア","エンカン","シラエヒ","アカヒ","クルマエヒ","カイルイ","サツキョ"
1190 DATA "コウケイ"
1200 FOR I=1 TO 38
1210 READ Item2$(I)
1220 NEXT I
1230 DATA "","","","","","","","","","","","","","","","","","",""
1240 DATA "","カント","","","","","","","","","","","","","",""
1250 DATA "","","カツオ","","","","","","","","","","","",""
1260 DATA "","","","","","","","","マズ","ワイカ","","ハツ"
1270 DATA "スルメイ","スルメイ","","","","","","","",""
1280 DATA "",""
1290 ! ----- チク -----
1300 FOR I=1 TO 9
1310 READ Chiku$(I)
1320 NEXT I
1330 DATA "クロハ","キョウテン","ウオツ","ナメリカワ","ミスハシ","イワセ"
1340 DATA "ヨカタ","シンミナト","ヒミ"
1350 ! ----- キョクヨウ -----
1360 FOR I=1 TO 4
1370 READ Gyogyo$(I)
1380 NEXT I
1390 DATA "テイチ","キョセン","ハツソウハリ","ソノタ"
1400 ! ----- FORMAT or IMAGE -----
1410 FOR I=1 TO 38
1420 READ Format1$(I)
1430 NEXT I
1440 DATA "7A,#", "8A,#", "6A,#", "6A,#", "6A,#", "6A,#", "8A,#", "7A,#", "6A,#"
1450 DATA "6A,#", "6A,#", "5A,#", "5A,#", "5A,#", "5A,#", "5A,#", "6A,#", "5A,#"
1460 DATA "5A,#", "5A,#", "5A,#", "6A,#", "5A,#", "5A,#", "6A,#", "7A,#", "5A,#"
1470 DATA "4A,#", "6A,#", "5A,#", "6A,#", "6A,#", "6A,#", "6A,#", "7A,#", "5A,#"
1480 DATA "7A,#", "8A,#"
1490 FOR I=1 TO 38
1500 READ Format2$(I)
1510 NEXT I
1520 DATA "7D,#", "8D,#", "6D,#", "6D,#", "6D,#", "6D,#", "8D,#", "7D,#", "6D,#"
1530 DATA "6D,#", "6D,#", "5D,#", "5D,#", "5D,#", "5D,#", "6D,#", "6D,#", "5D,#"
1540 DATA "5D,#", "5D,#", "5D,#", "6D,#", "5D,#", "5D,#", "6D,#", "7D,#", "5D,#"
1550 DATA "4D,#", "6D,#", "5D,#", "6D,#", "6D,#", "6D,#", "6D,#", "7D,#", "5D,#"
1560 DATA "7D,#", "8D,#"
1570 Menu: ! ----- メニュー ルーチン -----
1580 DISP "シコトノ シテイ、ソフトキ ヨ オセ"
1590 ON KEY 0 LABEL "ニューリョク" GOSUB Input

```

付表1 つづき①

```

1600 ON KEY 1 LABEL "フプリント" GOSUB Output
1610 ON KEY 2 LABEL "ディスクニカク" GOSUB Diskout
1620 ON KEY 3 LABEL "ディスクカラヨク" GOSUB Diskin
1630 ON KEY 4 LABEL "メモリクリア" GOSUB Mclear
1640 ON KEY 5 LABEL "オワリ" GOSUB End
1650 ON KEY 6 LABEL "9825カラヨク" GOSUB Hp9825in
1660 GOTO 1580
1670 Mclear: ! ***** メモリ クリアー サフ Mclear *****
1680 MAT Data1= (0) ! 0レツ ノ カクヨウヨク ヲ スハテ ヒロニスル
1690 PRINT "メモリ クリアー オワリ"
1700 RETURN
1710 Input: ! ***** ニュウリョク サフ ムーチン Input *****
1720 Printer_no=1
1730 PRINTER IS Printer_no ! CRT ニ シュツリョク スル
1740 PRINT CHR$(128) ! Inverse mode カイシヨ
1750 GOSUB Set ! テン,ツキ,シユン,フアイル,マイヨ セツテイ スル
1760 GOSUB Set1 ! キョウ ト レツ ヲ セツテイ スル
1770 IF Start_gyo=0 THEN GOTO 2720 ! 0シメノ キョウ カ 0 ナラ ニュウ ムーチン ハトフ
1780 Page=INT((Start_retsu-3)/9)+1 ! ヒョウシメ スル ハロシメ ノ ケイザン
1790 IF Start_retsu=1 OR Start_retsu=2 THEN Page=1
1800 Page=Page+(INT((Start_gyo-1)/10)+1)*10
1810 Star=-2 ! ----- 97 ニュウリョク -----
1820 GOSUB Crti ! ニュウリョク ヨウ ノ ヒョウ ヲ CRT ニ ヒョウシメ スル
1830 I=Start_gyo
1840 J=Start_retsu
1850 Datain: ! -----
1860 IF I<11 THEN Smark_gyo=6 ! ウイジンブツン ノ ヒョウ ノ カール ノ ヲテホクコウ ノ イチ
1870 IF I>10 THEN Smark_gyo=-4 ! シタジンブツン ノ ヒョウ ノ カール ノ ヲコホクコウ ノ イチ
1880 Page_next=INT((J-3)/9)+1 ! I,J カハ ハンカシタリテ アタメテ ハロシメ ヲ ケイザンスル
1890 IF J=1 OR J=2 THEN Page_next=1
1900 Page_next=Page_next+(INT((I-1)/10)+1)*10
1910 IF Page_next>Page OR Page_next<Page THEN ! ハロシメ 0シメノコウ カ カワツタラ
1920 Page=Page_next
1930 GOSUB Crti ! アタラシイ ハロシメ ヲ ヒョウシメ スル
1940 END IF
1950 Smark_retsu=0 ! カール ノ ヲコホクコウ ノ イチ ノ ケイザン
1960 FOR K=9*(Page-(10*INT(Page/10))-1)+3 TO J-1
1970 Mark_retsu%=Format2$(K)[1,1]
1980 Smark_retsu=Smark_retsu+VAL(Mark_retsu%)
1990 NEXT K
2000 Smark_retsu=Smark_retsu+16
2010 IF J=1 THEN Smark_retsu=1
2020 IF J=2 THEN Smark_retsu=8
2030 CONTROL 1;Smark_retsu,I+Smark_gyo ! カール ノ サヒョウ ヲ シテイスル
2040 PRINT CHR$(129); ! Inverse mode ニ スル
2050 PRINT USING Format2$(J);Data1(I,J) ! ステニ 0レツテイヒル テーグ ヲ ヒョウシメ
2060 Data=Data1(I,J)
2070 IF J=1 THEN INPUT "キョキョウ? (テイシ=1,キョセン=2,0ツウ00リ=3,ソノ9=4,チュウシ=-9)",Data
2080 IF J=2 THEN INPUT "チク?(クハ=1,キョウテン=2,ウオス=3,ナメリカワ=4,ミヌ0シ=5,イワヒ=6,ヨカタ=7,
シンミナト=8,ヒミ=9,チュウシ=-9)",Data
2100 IF J>2 THEN INPUT "テーグ? (ヒョウリ=-4,ミキ=-6,ウイ=-8,シタ=-2,テイシ=-5,チュウシ=-9)",Data
2110 CONTROL 1;Smark_retsu,I+Smark_gyo
2120 PRINT CHR$(128); ! Inverse mode カイシヨ
2130 PRINT USING Format2$(J);Data1(I,J) ! ニュウリョク シタ テーグ ヲ ヒョウシメ スル
2140 IF Data=-9 THEN GOTO Total
2150 IF Data<0 THEN ! マイナス ノ アタイ カ ニュウリョク サレタラ
2160 Star=Data ! カール ノ イトウホクコウ ヲ カエル
2170 ELSE
2180 Data1(I,J)=Data
2190 END IF

```

付表1 つづき②

```

2200 CONTROL 1:Smark_retsu,I+Smark_gyo
2210 IF J=1 THEN PRINT USING Format2$(1);Data1(I,1)
2220 IF J=2 THEN PRINT USING Format2$(2);Data1(I,2)
2230 IF J>2 THEN PRINT USING Format2$(J);Data1(I,J)
2240 IF Star=-4 THEN J=J-1          ! カーソル ヲ ヒタリ ハ
2250 IF Star=-6 THEN J=J+1        !          ミキ ハ
2260 IF Star=-8 THEN I=I-1        !          ウエ ハ
2270 IF Star=-2 THEN I=I+1        !          シタ ハ ウコカス
2280 IF Star=-5 THEN              ! カーソル 01 テイシ
2290   I=I
2300   J=J
2310 END IF
2320 IF I>End_gyo THEN             ! カーソル カ カタン ニ イツラ
2330   I=Start_gyo                ! ショウタン ニ モトシ
2340   J=J+1                      ! 1レツ ミキニ ウコカス
2350   IF J>End_retsu THEN J=Start_retsu ! ミキ01シ ニ イツラ ヒタリ01シ ハ
2360 END IF
2370 IF I<Start_gyo THEN
2380   I=End_gyo
2390   J=J-1
2400   IF J<Start_retsu THEN J=Start_retsu
2410 END IF
2420 IF J>End_retsu THEN
2430   J=Start_retsu
2440   I=I+1
2450   IF I>End_gyo THEN I=Start_gyo
2460 END IF
2470 IF J<Start_retsu THEN
2480   J=End_retsu
2490   I=I-1
2500   IF I<Start_gyo THEN I=Start_gyo
2510 END IF
2520 GOTO Dain
2530 Total: !----- コウケイ ノ ケイサン ト ヒョウシ -----
2540 FOR K=1 TO 20
2550   Yokokei=0
2560   FOR L=4 TO 37
2570     Yokokei=Yokokei+Data1(K,L)
2580     NEXT L
2590     Data1(K,38)=Yokokei
2600   NEXT K
2610 Start_gyo=1
2620 End_gyo=20
2630 Page=14
2640 GOSUB Crt1                    ! ショウタン ノ ミキ01シ ノ ヘーシ
2650 DISP "チェック カオウツラ CONTINUE KEY ヲ オセ" ! ヲ ヒョウシスル
2660 PAUSE ! フウクラム ノ ショウコウ ヲ イチシテイシスル,CONTINUE キー ヲ オスコト ニ ヨリ ソウコウ スル,
2670 Page=24                       ! カタン ノ ミキ01シ ノ ヘーシ
2680 GOSUB Crt1                    ! ヲ ヒョウシスル
2690 DISP "チェック カオウツラ CONTINUE KEY ヲ オセ"
2700 PAUSE
2710 GOTO 1760
2720 RETURN
2730 Crt1: !***** TV マチノ フォリナ ハ ヒョウ ヲ ヒョウシ スル サブルーチン Crt1 *****
2740 Jun$(1)="ショウ"
2750 Jun$(2)=" チュウ"
2760 Jun$(3)=" ケ"
2770 IF Printer no=1 THEN          ! CRT ニ シュツリョク スルトキ01
2780   Crt_gyo1=10*(INT(Page/10))-9 ! フンカツヒョウ ニ スルノテ
2790   Crt_gyo2=Crt_gyo1+9        ! キョウ ト

```


付表1 つづき③

```

2800     Crt_retsu1=9*(Page-10*(INT(Page/10))-1)+3      ! レツ ヲ
2810     Crt_retsu2=Crt_retsu1+8                        ! シテイ スル
2820     OUTPUT 2 USING " ",K";CHR$(255)&"K"          ! カクメン ノ ショウキョ
2830     ELSE                                           ! Printer ニ シュツリョク スルトキハ
2840     Crt_gyo1=Start_gyo                             ! ヒョウ センタイ テアラ
2850     Crt_gyo2=End_gyo
2860     Crt_retsu1=3
2870     Crt_retsu2=38
2880     END IF
2890     PRINTER IS Printer_no                          ! シュツリョクキキ ヲ シテイスル
2900     IF Printer_no=701 THEN PRINT CHR$(27);"&k0S"  ! フツウ モシモ モートニスル
2910     Fmt1:IMAGE " トヤマケン ",4D,"ネン",3D,"カツ ",4A,"シユン キョキョウ テーグ(100KG) ファイル名="
2920     PRINT USING Fmt1;Year,Month,Jun$(Jun),Filename$ ! Fmt1 ノ Image テア フォリント
2930     PRINT
2940     IF Printer_no=701 THEN PRINT CHR$(27);"&k2S"  ! アツシユク モシモ モートニスル
2950     PRINT USING Format1$(1);Item1$(1)              ! キョキョウ ト フォリントスル
2960     PRINT USING Format1$(2);Item1$(2)              ! チク ト フォリントスル
2970     FOR Jj=Crt_retsu1 TO Crt_retsu2
2980         PRINT USING Format1$(Jj);Item1$(Jj)        ! キョシユメイ ヲ フォリント
2990     NEXT Jj
3000     PRINT
3010     PRINT USING Format1$(1);Item2$(1)
3020     PRINT USING Format1$(2);Item2$(2)
3030     FOR Jj=Crt_retsu1 TO Crt_retsu2
3040         PRINT USING Format1$(Jj);Item2$(Jj)        ! キョシユメイ ノ 2キョウメイ ヲ フォリント
3050     NEXT Jj
3060     PRINT
3070     PRINT USING Format2$(1);1                       ! レツ ハンコウ ヲ フォリント
3080     PRINT USING Format2$(2);2
3090     FOR Jj=Crt_retsu1 TO Crt_retsu2
3100         PRINT USING Format2$(Jj);Jj
3110     NEXT Jj
3120     PRINT
3130     PRINT
3140     FOR Ii=Crt_gyo1 TO Crt_gyo2
3150         IF Data1(Ii,1)<1 OR Data1(Ii,1)>4 THEN    ! キョキョウコートカ 1-4 テナイトキハ
3160             PRINT USING Format2$(1);Data1(Ii,1)    ! コートキ ヲ フォリント
3170         ELSE                                       ! 1-4 ノ トキハ
3180             PRINT USING Format1$(1);Gyogyo$(Data1(Ii,1)) ! キョキョウメイ ヲ フォリント
3190         END IF
3200         IF Data1(Ii,2)<1 OR Data1(Ii,2)>9 THEN    ! チクコートカ 1-9 テナイトキハ
3210             PRINT USING Format2$(2);Data1(Ii,2)    ! コートキ ヲ フォリント
3220         ELSE                                       ! 1-9 ノ トキハ
3230             PRINT USING Format1$(2);Chiku$(Data1(Ii,2)) ! チクメイ ヲ フォリント
3240         END IF
3250         FOR Jj=Crt_retsu1 TO Crt_retsu2
3260             PRINT USING Format2$(Jj);Data1(Ii,Jj)  ! キョカクリョウ ヲ フォリント
3270         NEXT Jj
3280         PRINT
3290     NEXT Ii
3300     RETURN
3310     Set: ! ***** ネン,ツキ,シユン ノ セット ワフメーチン Set *****
3320     INPUT "セイレキ ネン",Year
3330     PRINT "セイレキ ネン=",Year
3340     INPUT "ツキ",Month
3350     PRINT "ツキ=",Month
3360     INPUT "シユン",Jun
3370     PRINT "シユン",Jun
3380     Filename$="TY0000E "
3390     Filename$(3,4)=VAL$(Year-1900)

```

付表1 つづき④

```

3400 IF Month>=10 THEN
3410   Filename$[5,6]=VAL$(Month)
3420 ELSE
3430   Filename$[6,6]=VAL$(Month)
3440 END IF
3450 IF Jun=1 THEN Filename$[7,7]="E"
3460 IF Jun=2 THEN Filename$[7,7]="M"
3470 IF Jun=3 THEN Filename$[7,7]="L"
3480 RETURN
3490 Output: ! ***** フォント サブルーチン Output *****
3500 PRINTER IS CRT
3510 Printer_no=701
3520 GOSUB Set
3530 GOSUB Set1
3540 IF Start_gyo=0 THEN GOTO 3580
3550 GOSUB Cr1
3560 PRINT CHR$(27);"&k0S"           ! フツウ モシモ モート ニ スル
3570 PRINTER IS CRT
3580 RETURN
3590 Set1: ! ***** キョウ, レツ ノ セット サブルーチン Set1 *****
3600 INPUT "ハシメノキョウ 1-20, Xニユ=0", Start_gyo
3610 PRINT "ハシメノキョウ =", Start_gyo
3620 IF Start_gyo=0 THEN GOTO 4040
3630 IF Start_gyo<0 OR Start_gyo>20 THEN
3640   BEEP 300,.2           ! 300 Hz ノ BEEP オンヲ 0.2 ヒョウダス
3650   PRINT "ハシカキイ イレナシ"
3660   GOTO 3600
3670 END IF
3680 INPUT "オウリノキョウ 1-20", End_gyo
3690 PRINT "オウリノキョウ =", End_gyo
3700 IF End_gyo<Start_gyo THEN
3710   BEEP 300,.2
3720   PRINT "ハシメノキョウヨリチイサイ, イレナシ"
3730   GOTO 3680
3740 END IF
3750 IF End_gyo>20 THEN
3760   BEEP 300,.2
3770   PRINT "ハシカキイ イレナシ"
3780   GOTO 3680
3790 END IF
3800 IF Printer_no=1 THEN
3810   INPUT "ハシメノレツ 1-37", Start_retsu
3820   PRINT "ハシメノレツ =", Start_retsu
3830   IF Start_retsu<1 OR Start_retsu>37 THEN
3840     BEEP 300,.2
3850     PRINT "ハシカキイ イレナシ"
3860     GOTO 3810
3870   END IF
3880   INPUT "オウリノレツ 1-37", End_retsu
3890   PRINT "オウリノレツ = ", End_retsu
3900   IF End_retsu<Start_retsu THEN
3910     BEEP 300,.2
3920     PRINT "ハシメノレツヨリチイサイ, イレナシ"
3930     GOTO 3880
3940   END IF
3950   IF End_retsu>37 THEN
3960     BEEP 300,.2
3970     PRINT "ハシカキイ イレナシ"
3980     GOTO 3880
3990   END IF

```

付表1 つづき⑤

```

4000 ELSE
4010   Start_retsu=3
4020   End_retsu=38
4030 END IF
4040 RETURN
4050 Diskin: ! ***** ディスク カラ テーマヲヨム サブルーチン Diskin *****
4060 ON ERROR GOTO 4080
4070 GOTO 4110
4080 BEEP 300,.5
4090 PRINT "イレナオシ!"
4100 GOTO 4630
4110 MASS STORAGE IS ":INTERNAL,4,1" ! 1 ユニツ(ヒタリノ) ディスク トライフ ヲ シテイ スル。
4120 GOSUB Set
4130 PRINT "ファイル名=",Filename$
4140 INPUT "タマシイデスカ?(Y/N)",Yn1$
4150 IF Yn1$="Y" THEN
4160   GOTO 4220
4170 ELSE
4180   BEEP 300,.2
4190   DISP "イレナオシ"
4200   GOTO 4120
4210 END IF
4220 DISP "テーマ ディスク ヲ ヒタリノ トライフ ニ イレ CONTITUE キイ ヲ オセ!"
4230 PAUSE
4240 ASSIGN @Path1 TO Filename$ ! ディスク トライフ ニ ファイル ヲ ヒラク。
4250 ENTER @Path1:Data1(*) ! ユニツ ノ ナカニ ヲ マルコト ディスク カラ ニユクヨク スル。
4260 ASSIGN @Path1 TO * ! ファイル ヲ トシム。
4270 MASS STORAGE IS ":INTERNAL,4,0"
4280 RETURN
4290 Diskout: ! ***** ディスク ニ テーマヲカク サブルーチン Diskout *****
4300 ON ERROR GOTO 4320
4310 GOTO 4350
4320 BEEP 300,.5
4330 PRINT "イレナオシ!"
4340 GOTO 4620
4350 MASS STORAGE IS ":INTERNAL,4,1"
4360 GOSUB Set
4370 PRINT "ファイル名=",Filename$
4380 DISP "テーマ ディスク ヲ ヒタリノ トライフ ニ イレ CONTITUE キイ ヲ オセ!"
4390 PAUSE
4400 CAT;SELECT Filename$,COUNT File_nos,NO HEADER ! コノ ナマエノ ファイル カ アルカ シラヘム
4410 IF File_nos>0 THEN ! アレハ
4420   INPUT "ファイルカ ステンアル。ファイル ノ ナカニ ヲ カキナオシマスカ?(Y/N)",Yn1$
4430   IF Yn1$="Y" OR Yn1$="y" THEN
4440     GOTO 4590
4450   ELSE
4460     BEEP 300,.2
4470     PRINT "ファイル名 マダハ ディスクット ヲ タシカメテ イレナオセ!"
4480     GOTO 4360
4490   END IF
4500 END IF ! ナケレバ
4510 INPUT "アタラシイ ファイル ヲ ツクリマス。 ファイル名ハ タマシイカ?(Y/N)",Yn2$
4520 IF Yn2$="Y" OR Yn2$="y" THEN
4530   CREATE BDAT Filename$,800,8 ! ディスク ニ アタラシイ ファイル ヲ カクホ スル。
4540 ELSE
4550   BEEP 300,.2
4560   PRINT "イレナオシ"
4570   GOTO 4360
4580 END IF
4590 ASSIGN @Path1 TO Filename$ ! Path1 トイウ I/O ケイロ ヲ ファイル ニ ワリアテル

```

付表1 つづき⑥

```

4600 OUTPUT @Path1:Data1(*)           ! 0-イレツ ノ ナカミ ヲ マルコト ティスクニ カク
4610 ASSIGN @Path1 TO *                ! I/O ケイロ ヲ トシム
4620 MASS STORAGE IS ":INTERNAL,4,0"  ! 000ン(ミキカワ) ノ ティスク トライフ ヲ シテイスル
4630 RETURN
4640 Hp9825in: ! ***** HP9825 カラ テータ ヲ ヨク サフル-チン Hp9825in *****
4650 PRINT "9825 カラ ノ ニュウリョク"
4660 FOR I=1 TO 20
4670     FOR J=1 TO 38
4680         ENTER 707;Data1(I,J)      ! 70700ン ノ キキ(HP9825A) カラ テータ ヲ ニュウリョク
4690     NEXT J
4700 NEXT I
4710 PRINT "9825 カラ ノ ニュウリョク オウリ"
4720 RETURN
4730 End: ! ***** エント サブ END *****
4740 DISP "オウリ, コックロウサマ!! "
4750 END
    
```

付表2 出力見本

トママケン 1984年 10カハツ ショウシユン キョキョウ テータ(100KG) ファイル,メイ=TY8410E

サマサマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ	サマ
1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	
サマ	サマ	66	34	0	0	0	0	0	72	0	0	0	0	2	137	5	0	
サマ	サマ	40	0	0	0	0	0	0	16	0	0	0	0	0	27	0	0	
サマ	サマ	40	10	0	0	1	4	0	48	0	0	0	0	15	154	0	0	
サマ	サマ	10	0	0	0	0	0	0	5	0	0	0	0	0	68	0	0	
サマ	サマ	30	0	0	0	0	0	0	1	0	0	0	0	1	160	0	0	
サマ	サマ	9	0	0	0	0	0	0	44	0	0	0	1	4	66	0	0	
サマ	サマ	45	1	0	2	0	1	0	26	0	0	0	0	0	129	0	6	
サマ	サマ	108	0	62	130	14	142	0	12	0	0	0	0	4	612	53	0	
サマ	サマ	126	2	65	14	19	57	0	83	0	0	0	8	6	806	417	0	
サマ	サマ	34	0	155	0	172	71	0	60	0	0	0	0	0	530	14	102	
サマ	サマ	16	0	0	0	0	0	0	3	0	0	0	0	0	0	0	0	
サマ	サマ	152	0	0	0	0	0	0	0	0	0	0	0	0	26	0	0	
サマ	サマ	120	0	0	0	0	0	0	0	0	0	0	0	0	19	0	0	
サマ	サマ	109	0	0	0	0	0	0	0	0	0	0	0	1	6	0	0	
サマ	サマ	49	0	0	0	0	0	0	0	0	0	0	0	0	?	0	0	
サマ	サマ	91	0	0	0	0	0	0	0	0	0	0	0	0	0	0	2	
サマ	サマ	135	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	
サマ	サマ	58	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	
サマ	サマ	144	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	

515	77*	9397	*9140	ヒラメ	94	9140	ストウ997	9397	97	ハニス*	75	7774	エカ*ン	ラビ*	7011*	7671*	7414	9*77*3	7*974
19	20	21	22	23	24	25	26	27	28	29	30	31	32	33	34	35	36	37	38
5	2	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	23	280
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	43
22	7	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	261
19	2	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	2	96
1	2	1	0	0	1	0	0	0	0	0	0	0	0	0	0	0	0	2	169
12	5	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	2	134
0	0	0	0	0	0	0	0	0	2	0	0	0	0	0	0	0	0	8	175
50	7	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	45	1131
271	6	0	0	0	0	0	0	0	7	0	0	0	0	0	0	0	0	157	1918
2	0	0	0	0	0	0	0	0	1	0	0	0	1	0	0	0	0	5	1163
0	0	0	0	0	0	0	0	0	0	0	279	0	0	0	0	0	0	0	282
0	0	0	0	0	0	0	2	0	0	10	0	0	0	0	0	0	0	0	38
0	0	0	0	0	0	0	58	0	0	81	0	548	0	0	0	0	0	0	706
0	0	0	0	1	0	0	8	0	0	56	0	0	0	0	8	0	26	4	110
0	0	0	0	0	0	0	1	0	0	0	0	0	0	1	0	0	0	2	6
0	0	0	0	0	0	0	0	0	0	0	0	0	0	90	4	0	0	7	103
0	0	0	0	1	0	0	30	0	0	0	0	0	0	0	0	0	0	26	57
0	0	0	0	0	0	0	26	0	0	0	0	35	0	18	0	0	0	0	79
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	52	52